ANALYTIC GEOMETRY OF THE PLANE AND MATHEMATICA

MAJA BATOR, ZVONKO ČERIN AND MILENA ĆULAV, ZAGREB

Abstract. We describe the use of the program Mathematica in the analytic geometry of the plane in the rectangular coordinates. As an illustration of possible applications of this method we present the solutions of fifteen problems from the problem book for the first class of gymnasiums in Croatia.

In this article we describe the computer approach to the analytic geometry of the plane. In order to do this we shall use the symbolic computation program **Mathematica**. Of course, the same could be done in the rival program **Maple V**. These are the most widely known and the most popular extensive systems that "know mathematics". Each of them has its own programming language. Our task is reduced to describing basic functions that are needed for solving geometry problems with the analytic method.

This is the translation to English of the article [2] that is in Croatian. In the references [1], [3], [4] that are in Croatian the same task was done in the program Maple V. The whole project is the result of the second author's course "Geometry and computers" at the Mathematics Department of the University of Zagreb (in Croatia) in which the first and the third authors (the undergraduate students) have been enrolled in the academic year 2002/2003.

The key idea of the analytic geometry is to associate algebraic entities with geometric objects and then investigate them using algebraic methods. Since the simplest way to achieve this is in the plane we start our study there.

The building blocks in the plane are **points** so that we must first decide how to represent them by suitable algebraic objects. The selection of a coordinate system is the easiest method of doing this. Among several such known systems (e. g. rectangular, skew-angled, polar, trilinear, barycentric,...) we choose the rectangular coordinate system for its simplicity and familiarity.

This system is determined by two perpendicular oriented lines p_x and p_y and by the point E different from their intersection $O = p_x \cap p_y$ selected on one of them (say on p_x). The point O is called the *origin*, the lines p_x and p_y are the *x-axis* and the *y-axis*. The point E is the *unit point* (see Figure 1).



FIGURE 1. The rectangular coordinate system.

The position of any point T in the plane is completely determined by the ordered pair (x, y) of real numbers which are named the (rectangular) *coordinates* of the point T. The numbers x and y are the following ratios of oriented distances

$$x = \frac{|OT_x|}{|OE|}, \qquad y = \frac{|OT_y|}{|OE|},$$

where T_x and T_y are the projections of the point T on the x-axis and on the y-axis.

The input of points on the plane in Mathematica is quite simple because they are just ordered pairs of real numbers. For example, the input

```
tA:={2, 3}; tB:={5, 7}; tC:={-2, 0}; tT:={x, y};
```

defines four points on the plane $A(2,3)$, $B(5,7)$, $C(-2,0)$, $T(x,y)$.

Now, when we know how to define points on the plane let us measure *distances* between them.

For points A and B on the plane with coordinates (a, u) and (b, v) the distance $|AB|$ is the hypothenuse of the right triangle ABC, where C is the intersection of the parallel to the x-axis through A and the parallel to the y-axis through B. The legs of that triangle are (horizontal) $|b-a|$ and (vertical) $|u-v|$ (see Figure 2). Therefore, by Pythagorean theorem, the distance is

$$|AB| = \sqrt{(b-a)^2 + (v-u)^2}.$$

The following function calculates distances in Mathematica:

```
distance[{a_,u_}, {b_,v_}]:=Sqrt[FS[(b-a)^2+(v-u)^2]]
```

The name of this function is *distance*. It asks for two ordered pairs of real numbers. The first pair has the components a and u while the



FIGURE 2. The distance between the points $A(a,\ u)$ and $B(b,\ v)$ is $|AB| = \sqrt{(b-a)^2 + (v-u)^2}$.

components of the second pair are b and v. The machine first computes $(b-a)^2 + (v-u)^2$ and then tries as much as possible to simplify and factor this sum of squares (the command `FS`). In the end it finds the square root of everything (the command `Sqrt`).

In order to cut down typing we introduce the shortening `FS` for the simultaneous use of commands `Factor` and `FullSimplify` that will be used frequently.

```
FS[m_]:=Factor[FullSimplify[m]];
```

Many times it is important to determine the *midpoint* of the segment whose endpoints are given. From the Figure 3 below we infer that the coordinates of the midpoint P of the segment with endpoints $A(a,\ u)$ and $B(b,\ v)$ are $\frac{a+b}{2}$ and $\frac{u+v}{2}$.

The function which computes the midpoint of a segment has in the program Mathematica the following form:

```
midpoint[{a_,u_}, {b_,v_}] := FS[{(a+b)/2,(u+v)/2}]
```

It is slightly harder to find the coordinates of the point which divides a segment in a given ratio. In other words, we are looking for the point C on the line AB such that the ratio of lengths of oriented segments $\dfrac{|AC|}{|CB|}$ is equal to a given real number λ (that must be different from -1). Its coordinates are $\dfrac{a+\lambda\, b}{1+\lambda}$ (abscissa) and $\dfrac{u+\lambda\, v}{1+\lambda}$ (ordinate). The number λ is often given as the quotient $\frac{m}{n}$ of integers m and n (that are not opposite to each other).

The corresponding functions are defined in Mathematica as follows:



FIGURE 3. The coordinates of the midpoint P of the segment $\overline{AB}$ are the arithmetic mean of the coordinates of its endpoints.

```
ratio[{a_,u_},{b_,v_},k_]:=
            FS[{(a+k*b)/(1+k),(u+k*v)/(1+k)}]
```

```
ratiomn[{a_,u_}, {b_,v_},m_,n_] :=
            FS[{(a*n+b*m)/(m+n),(u*n+v*m)/(m+n)}]
```

Of course, the midpoint function is a special case of these functions. We get the midpoint for $\lambda = 1$ and for $m = 1$ and $n = 1$.

Besides points the most important objects in the plane are **lines**. Recall that lines are represented by linear equations. More precisely, if a, b and c are real numbers and a and b are not both zero, then the set of all points $P(x, y)$ in the plane whose coordinates satisfy the equation $a\,x + b\,y + c = 0$ is a line. Conversely, for every line in the plane we can find such real numbers a, b and c so that each of its points satisfies the above equation.

This explains why we shall represent lines in Mathematica as ordered triples $\{a, b, c\}$ of coefficients of their linear equations. For example, the input

```
pX:={1, 0, 0}; pY:={0, 1, 0}; pD:={1, -1, 0}; pG:={-1, 2, 2}
```

defines four lines in the plane. They are the x-axis, the y-axis, the bisector of the first and the third quadrant and the line $-x + 2\,y + 2 = 0$.

The line is usually determined by one of its points $P_0(x_0, y_0)$ and the tangent k of the angle which it makes with the positive direction of the x-axis. In this case its equation is

$$y - y_0 = k\,(x - x_0),$$

so that in Mathematica we use for it the following function:

```
line1[k_,{b1_,b2_}] := FS[{k,-1,b2-b1*k}];
```

The second most common method of representing a line is when we know two different points $P_1(x_1,\, y_1)$ and $P_2(x_2,\, y_2)$ which lie on it. The equation of the line is

$$y - y_1 = \frac{y_2 - y_1}{x_2 - x_1}\,(x - x_1)$$

when $x_1 \neq x_2$. Rewriting this equation in the form

$$(y_2 - y_1)\,x + (x_1 - x_2)\,y + (x_1\,y_2 - x_2\,y_1) = 0$$

we see that it holds also when $x_1 = x_2$. Therefore, in this situation the corresponding Mathematica function is the following:

```
line2[{x1_,y1_},{x2_,y2_}] :=FS[{y2-y1,x1-x2,x1*y2-x2*y1}];
```

Sometimes it is useful to have the following function which investigates whether the given point lies on the given line. The letter Q in its name suggests the word "question" and is common for Mathematica. The point is on the line if and only if the function evaluates to zero.

```
onlineQ[{a_,b_},{x_,y_,z_}] := FS[a*x+y*b+z];
```

It is now easy to find the condition that the coordinates of three points $P_1(x_1,\, y_1)$, $P_2(x_2,\, y_2)$, and $P_3(x_3,\, y_3)$ must satisfy in order that they lie on a line (i. e., that they are collinear). In fact, by now we know the equation of the line through the points P_1 and P_2. Since the point P_3 must lie on this line its coordinates must satisfy this equation. In other words,

```
onlineQ[{x3,y3},line2[{x1,y1},{x2,y2}]]
```

is zero (i. e., $(y_2 - y_1)\,x_3 + (x_1 - x_2)\,y_3 + (x_1\,y_2 - x_2\,y_1) = 0$).

This explains the definition of the following Mathematica function which tests if three given points are collinear.

```
collinearQ[{x1_,y1_},{x2_,y2_},{x3_,y3_}] :=
                    FS[y2*x3-y1*x3+x1*y3-x2*y3+x1*y2-x2*y1]
```

Similarly as in the case of the function `onlineQ`, three points in the plane will be collinear if and only if the function `collinearQ` evaluates in them to zero.

Different lines $A_1\,x + B_1\,y + C_1 = 0$ and $A_2\,x + B_2\,y + C_2 = 0$ in the plane either intersect or are parallel. If they intersect, then the coordinates of the point of their intersection are

$$x = \frac{B_1\,C_2 - B_2\,C_1}{A_1\,B_2 - A_2\,B_1}, \qquad y = \frac{C_1\,A_2 - C_2\,A_1}{A_1\,B_2 - A_2\,B_1}.$$

We obtain them as output of the following Mathematica input:

```
Solve[{A1*x+B1*y+C1==0, A2*x+B2*y+C2==0}, {x, y}]
```

The assumption that the given lines intersect implies that the denominators in the above expressions are different from zero. Parallel lines do not have the intersection (in the finite plane) and $A_1\,B_2 - A_2\,B_1 = 0$ is the condition on their coefficients that must hold in order that they are parallel. Hence, the following Mathematica function determines the intersection of two lines in case when they are not parallel:

```
inter[{a_,b_,c_},{i_,j_,k_}] :=
     FS[(-j*c+k*b)/(-i*b+a*j), (i*c-a*k)/(-i*b+a*j)];
```

They will be parallel provided the use of the function `inter` results in the Mathematica error message of division with zero:

`Power::"infy" : "Infinite expression` $\frac{1}{0}$ `encountered."`

When the lines p_1 and p_2 with equations

$$A_1\,x + B_1\,y + C_1 = 0 \quad \text{and} \quad A_2\,x + B_2\,y + C_2 = 0$$

intersect, then the tangent of the angle φ among them is given by the formula

$$\tan\varphi = \frac{A_1B_2 - A_2B_1}{A_1A_2 + B_1B_2} = \frac{k_2 - k_1}{1 + k_1k_2},$$

where $k_1 = -\dfrac{A_1}{B_1}$ and $k_2 = -\dfrac{A_2}{B_2}$ are tangents of the angles φ_1 and φ_2 that they make with the positive direction of the x-axis (see Figure 4).



FIGURE 4. The angle among two intersecting lines.

The following function accomplishes this in Mathematica:

```
angle[{a_,b_,c_},{i_,j_,k_}] := FS[(a*j-i*b)/(a*i+b*j)];
```

Notice that the lines will be perpendicular when the use of this function in Mathematica reports the division with zero error.

It follows that the lines p_1 and p_2 are parallel if the angle among them is zero or if $A_1B_2 - A_2B_1 = 0$ or $k_1 = k_2$. On the other hand, they will be perpendicular if $A_1A_2 + B_1B_2 = 0$ or $k_2 = -\dfrac{1}{k_1}$ (the denominator of the above quotient is zero so that the angle is right).

These remarks justify the following definitions of Mathematica functions that give the parallel through a given point to a given line and the perpendicular through a given point to a given line. Similar functions test if two given lines are parallel or perpendicular:

```
parallel[{a_,b_},{x_,y_,z_}] :=FS[{x,y,-x*a-b*y}];
perpen[{a_,b_},{x_,y_,z_}] :=FS[{y,-x,x*b-y*a}];
parallelQ[{a_,b_,c_},{x_,y_,z_}] :=FS[a*y-x*b];
perpenQ[{a_,b_,c_},{x_,y_,z_}] :=FS[a*x+y*b];
```

When the functions `parallelQ` or `perpenQ`, for a given pair of lines, return the value zero, then these two lines are parallel or perpendicular, respectively.

For three different lines p_1, p_2 and p_3 in the plane with equations

$$A_1\,x + B_1\,y + C_1 = 0, \quad A_2\,x + B_2\,y + C_2 = 0, \quad A_3\,x + B_3\,y + C_3 = 0$$

we shall now determine the condition which their coefficients must satisfy in order that they are concurrent (i. e., that they are parallel or intersect in a point).

If the lines p_1 and p_2 intersect at the point $T(x,\,y)$ we know that the coordinates x and y are the quotients $\frac{B_1\,C_2 - B_2\,C_1}{A_1\,B_2 - A_2\,B_1}$ and $\frac{C_1\,A_2 - C_2\,A_1}{A_1\,B_2 - A_2\,B_1}$. Substituting these values into the equation of p_3 we get

$$\frac{A_3B_1C_2 - A_3B_2C_1 + B_3C_1A_2 - B_3C_2A_1 + C_3A_1B_2 - C_3A_2B_1}{A_1B_2 - A_2B_1} = 0.$$

Hence, these three lines have a common point provided the expression in the numerator of the left hand side

$$\Delta = A_3\,(B_1C_2 - B_2C_1) + B_3\,(C_1A_2 - C_2A_1) + C_3\,(A_1B_2 - A_2B_1)$$

is equal to zero.

If the lines p_1, p_2 and p_3 are parallel then their coefficients satisfy the relations $A_2 = \lambda A_1$, $B_2 = \lambda B_1$, $A_3 = \mu A_1$, $B_3 = \mu B_1$, for some real numbers λ and μ different from zero. When we substitute these values into Δ it is again equal to zero.

Conversely, if $\Delta = 0$, then after dividing both sides of this equality by $A_1B_2 - A_2B_1$ we infer that the intersection $T\left(\frac{B_1\,C_2 - B_2\,C_1}{A_1\,B_2 - A_2\,B_1}, \frac{C_1\,A_2 - C_2\,A_1}{A_1\,B_2 - A_2\,B_1}\right)$ of the lines p_1 and p_2 satisfies the equation of p_3. Hence, these three lines are concurrent. This argument holds only for $A_1\,B_2 - A_2\,B_1 \neq 0$, i. e., if the lines p_1 and p_2 are not parallel.

If the lines p_1 and p_2 are parallel, then there is a real number $\lambda \neq 0$ such that $A_2 = \lambda A_1$ and $B_2 = \lambda B_1$. Substituting these into the equality $\Delta = 0$ we obtain

$$(A_1 B_3 - A_3 B_1)(\lambda C_1 - C_2) = 0.$$

In the product on the left hand side the second parenthesis is different from zero because $p_1 \neq p_2$. Hence, $A_1 B_3 - A_3 B_1 = 0$ which means that the lines p_1 and p_3 are also parallel and the proof is completed because then the lines p_1, p_2 and p_3 are parallel.

In this way we explained how to define the following function in Mathematica which will test if three given lines are concurrent:

```
concurQ[{a_,b_,c_},{i_,j_,k_},{p_,q_,r_}] :=
              FS[a*j*r-a*k*q-i*b*r+i*c*q+p*b*k-p*c*j];
```

Hence, three lines either intersect in a point or are parallel provided the value of the function `concurQ` in them is zero.



Figure 5. The reflection of a point with respect to a point (left) and with respect to a line (right).

In solving problems using the analytic geometry it is often necessary to determine the projection of a point onto a line. Since the projection is the intersection of the line and the perpendicular to the line through the point, if we input into Mathematica

```
P:={p, q}; m:={a, b, c}; Q:=inter[m, perpen[P, m]];
```

the output will be the coordinates of the projection Q of the point P onto the line m. Hence, the corresponding function looks as follows:

```
project[{p_, q_},{a_, b_, c_}]:=
             FS[{-(c*a+b*q*a-p*b^2)/(b^2+a^2),
               (-b*c+q*a^2-a*p*b)/(b^2+a^2)}];
```

Similarly, many times we also need the reflection of a point with respect to a given point or a given line. When the point P is reflected with respect to the point Q the coordinates of the reflected point R can be determined using the function `ratio` for $\lambda = -2$ because the point R divides the segment $\overline{PQ}$ in the ratio -2.

```
P:={p, q}; Q:={a, b}; R:=ratio[P, Q, -2]
```

It follows that the function which associates to a pair of given points P and Q the reflected point R in Mathematica is defined by the input

```
reflectP[{p_, q_},{a_, b_}]:= FS[{2*a-p, 2*b-q}];
```

The reflection of a point P with respect to a line m is analogous. This time the required point R is the reflection of the point P with respect to the projection Q of the point P onto the line m:

```
P:={p, q}; m:={a, b, c}; R:=ratio[P, project[P, m], -2];
```

Hence, the corresponding Mathematica function has the following form:

```
reflectL[{p_, q_},{a_, b_,c_}] :=
            FS[{(b^2*p-a^2*p-2*a*b*q-2*c*a)/(a^2+b^2),
              (a^2*q-b^2*q-2*a*b*p-2*c*b)/(a^2+b^2)}];
```

This concludes the listing of the most basic functions for the analytic geometry of the plane. In the rest of this paper we shall present fifteen geometry problems from the problem collection [6] and give detailed solutions of them in Mathematica. The collection is for the first year high school level (age 15–16) but some solutions require knowledge from the second and the third year.

Our first example is the problem 395 from the book [6] that reads as follows:

Problem 1. Prove that the area P of a triangle ABC with vertices in the points $A(x_1, y_1)$, $B(x_2, y_2)$ and $C(x_3, y_3)$ is given by the formula:

$$P = \frac{|x_1(y_2 - y_3) + x_2(y_3 - y_1) + x_3(y_1 - y_2)|}{2}$$

or

$$P = \frac{|y_1(x_2 - x_3) + y_2(x_3 - x_1) + y_3(x_1 - x_2)|}{2}.$$

Proof. Recall that the area of a triangle is a half of the product of lengths of any of its sides with the corresponding altitude. Hence, with the help of Mathematica functions introduced earlier, the area is easily computed as follows:

```
tA:={Subscript[x,1],Subscript[y,1]};
tB:={Subscript[x,2],Subscript[y,2]};
tC:={Subscript[x,3],Subscript[y,3]};
tD:=project[tC,line2[tA,tB]];
vP:=FS[distance[tA,tB]*distance[tC,tD]/2];
```



FIGURE 6. The area of the triangle ABC is half of the product of the length of the base $\overline{AB}$ and the length of the corresponding altitude $\overline{CD}$.

The output in Mathematica will be a rather complicated expression

$$\frac{1}{2}\sqrt{\frac{(-y_3x_1 + x_3y_1 + y_3x_2 - x_2y_1 - x_3y_2 + x_1y_2)^2}{{x_2}^2 - 2\,x_2x_1 + {x_1}^2 + {y_1}^2 - 2\,y_1y_2 + {y_2}^2}}\sqrt{{x_2}^2 - 2\,x_2x_1 + {x_1}^2 + {y_1}^2 - 2\,y_1y_2 + {y_2}^2}.$$

As the computer is just a machine and we have not explained the nature of symbols representing the coordinates of the vertices, it will not cancel out the denominator in the first square root with the second square root even though they are clearly identical. It also does not notice that the square root of the square in the numerator of the first square root is equal to the absolute value

$$|-y_3x_1 + x_3y_1 + y_3x_2 - x_2y_1 - x_3y_2 + x_1y_2|\,.$$

When we make these simplifications we shall obviously get the required formula. □

It is interesting to note that without the absolute value the above formula computes the oriented area of the triangle ABC. If this triangle is positively oriented, i. e., if the movement $ABCA$ is in the counterclockwise direction, then this real number will be positive and otherwise is negative. It will be zero if and only if the points A, B and C are collinear.

The function that gives this oriented area in Mathematica is realized in the following input:

```
area[{a_, x_},{b_, y_},{c_, z_}]:=
                FS[(x*c-b*x-a*z+a*y+b*z-c*y)/2];
```

The second example is the problem 425 from the same book [6].

Problem 2. Let ABC be a triangle and let U, V, W be midpoints of sides $\overline{BC}$, $\overline{CA}$ and $\overline{AB}$. The segments $\overline{AU}$, $\overline{BV}$ and $\overline{CW}$ are called the **medians** of the triangle ABC. Prove analytically that the three medians intersect in a point that we call the **centroid** of the triangle and that the centroid divides each median in the ratio $2:1$ counting from the vertex.



FIGURE 7. The triangle ABC and its medians $\overline{AU}$, $\overline{BV}$ and $\overline{CW}$ intersecting in the centroid G.

Proof. The proof on the computer, in Mathematica, begins by typing the following input:

```
tA :={Subscript[x,1],Subscript[y,1]};
tB :={Subscript[x,2], Subscript[y,2]};
tC :={Subscript[x,3],Subscript[y,3]};
tU:=midpoint[tB,tC]; tV:=midpoint[tC,tA];
tW:=midpoint[tA,tB]; concurQ[line2[tA,tU],
            line2[tB,tV],line2[tC,tW]];
```

In amazingly short time the computer will output the value zero which proves that the medians intersect in a point. The coordinates of this point are revealed with the commands:

```
tG := inter[line2[tA,tU], line2[tB,tV]];
```

The point G has the coordinates $\left(\frac{x_1+x_2+x_3}{3}, \frac{y_1+y_2+y_3}{3}\right)$ so that we can immediately write down the Mathematica function which associates the centroid to a triangle:

```
centroid[{a_,x_},{b_,y_},{c_,z_}]:=FS[(a+b+c)/3,(x+y+z)/3]
```

In order to prove the second claim of the problem we shall find the point that divides the median of the vertex A (i. e., the segment $\overline{AU}$)

in the ratio 2 : 1 counting from the vertex A and show that it coincides with the point G (the centroid of the triangle ABC). The same argument could be repeated for the medians of the vertices B and C.

```
tT:=ratiomn[tA, tU, 2, 1]; distance[tG, tT]
```

Since the returned value is zero, the points G and T coincide so that the proof of the problem is completed successfully. □

The third example is the problem 989 also from the collection [6].

Problem 3. Prove that the midpoints of sides and the feet of the altitudes of a triangle lie on the same circle.



FIGURE 8. The midpoints of sides and the feet of the altitudes of a triangle lie on the same circle.

Proof. Without loss of generality we can assume that the points A, B and C are selected in the plane so that their coordinates are $(0, 0)$, $(c, 0)$ and (u, v), where c, u and v are real numbers with c and v different from zero.

```
eA:={0, 0}; eB:={c, 0}; eC:={u, v};
```

Then we get the midpoints of the sides applying the function `midpoint`:

```
eAp:=midpoint[eB,eC]; eBp:=midpoint[eC,eA];
eCp:=midpoint[eA,eB];
```

The feet of the altitudes are the projections of the vertices onto the opposite sidelines:

```
eApp:=project[eA,line2[eB,eC]];
eBpp:=project[eB,line2[eC,eA]];
eCpp:=project[eC,line2[eA,eB]];
```

The center of the circle circumscribed to a triangle is the intersection of perpendicular bisectors of its sides. Hence, in our situation, the center S of the circle circumscribed to the triangle $A'B'C'$ with vertices in the midpoints of sides is defined as follows:

```
eS:=inter[perpen[midpoint[eBp,eCp],line2[eBp,eCp]],
          perpen[midpoint[eCp,eAp],line2[eCp,eAp]]]
```

Applying the same method to the triangle $A''B''C''$ with vertices at the feet of the altitudes we can find the center T of its circumscribed circle.

```
eT:=inter[perpen[midpoint[eBpp,eCpp],line2[eBpp,eCpp]],
          perpen[midpoint[eCpp,eApp],line2[eCpp,eApp]]]
```

After we type in the above commands the computer will output the coordinates of the points S and T. We see that they are equal, so that the points S and T coincide.

In order to complete the proof it remains still to prove that the radii of the circumcircles of the triangles $A'B'C'$ and $A''B''C''$ are equal. This is checked in Mathematica with the following input:

```
FS[distance[eS,eCp]-distance[eT,eCpp]]
```

Since the returned value is zero the proof is successfully accomplished.

With almost no effort we can now prove that the radius of the above circle (also known as the nine-point circle because it also goes through the midpoints of the segments joining vertices with the orthocenter) is equal to the half of the radius of the circle circumscribed to the triangle ABC. In order to check this using the same method as above we first find the coordinates of the center O of the circumcircle of ABC

```
eO:=inter[perpen[midpoint[eB,eC],line2[eB,eC]],
          perpen[midpoint[eC,eA],line2[eC,eA]]]
```

and request from Mathematica to compute the following:

```
FS[distance[eO,eC]/distance[eS,eCp]]
```

Of course, the result is the number two. □

The fourth example are the problems 719 and 720 from the book [6].

Problem 4. Prove that if a triangle has two equal altitudes or two equal medians, then it is isosceles.

Proof. With the assumptions and the notation from the proof of the Problem 3, typing in

```
FS[distance[eA,eApp]^2-distance[eB,eBpp]^2]
```

we obtain $\frac{c^3v^2(2\,u-c)}{(v^2+u^2-2\,uc+c^2)(v^2+u^2)}$. Hence, if the altitudes AA'' and BB'' have the same lengths then $u = \frac{c}{2}$ so that ABC is an isosceles triangle because the vertex C lies on the perpendicular bisector of the side AB.

Similarly we see that after typing into the program Mathematica



FIGURE 9. Relations $|AA'| = |BB'|$ or $|AA''| = |BB''|$ imply $|BC| = |CA|$.

```
FS[distance[eA,eAp]^2-distance[eB,eBp]^2]
```

the output is $\frac{3\,c(2\,u-c)}{4}$ that leads to the same conclusion for medians. $\square$

More complicated to prove is the Problem 721 from [6]. Our method of its proof assumes the knowledge of the trigonometric functions (the cotangent in particular).

Problem 5. Prove that a triangle is isosceles if and only if it has two equal angle bisectors.



FIGURE 10. The sides BC and CA are equal if and only if the bisectors AA_i and BB_i of angles A and B are equal.

Proof. In order to have simple expressions we shall assume that the vertices A and B and the incenter I (i. e., the center I of the circle inscribed to the triangle ABC) have the coordinates $(0,\,0)$, $(f+g,\,0)$, and $(f,\,1)$, where f and g are positive real numbers. In fact, these are the cotangents of the halves of the angles A and B. In addition, we assumed that the inradius is equal to 1.

```
tA:={0, 0}; tB:={f+g, 0}; tI:={f, 1}; tJc:={f, 0};
```

If the points J_a, J_b, J_c are the projections of the incenter I onto the sides of ABC, then J_c has the coordinates $(f,\,0)$ while we get the coordinates of J_a as solutions of the following system of equations:

```
sys:=Solve[{distance[tB,{p, q}]==distance[tB,tJc],
         distance[tI,{p, q}]==1},{p, q}];
```

where p and q are the coordinates of the point J_a that we are trying to determine. This system has only two solutions. The first are the coordinates of the point J_c while the second are the required coordinates $\frac{f(g^2+1)+2\,g}{g^2+1}$ and $\frac{2\,g^2}{g^2+1}$ of the point J_a.

```
tJa:={p,q} /.  Extract[sys, 2]
```

In a similar way we can find also the coordinates $\frac{f(f^2-1)}{f^2+1}$ i $\frac{2\,f^2}{f^2+1}$ of the point J_b.

```
tJb:={p,q} /.  Extract[Solve[{distance[tA,{p, q}]==
     distance[tA,tJc], distance[tI,{p, q}]==1},{p, q}], 2]
```

Now we can find the points A_i and B_i of intersection of bisectors of angles A and B with the opposite sides as intersections $AI \cap BJ_a$ and $BI \cap AJ_b$.

```
tAi:=inter[line2[tA,tI],line2[tB,tJa]];
tBi:=inter[line2[tB,tI],line2[tA,tJb]];
```

Let us now ask the program Mathematica to calculate the difference of the squares of lengths of angle bisectors with the following input:

```
Q:=FS[distance[tA,tAi]^2-distance[tB,tBi]^2];
```

The output will be the quotient

$$\frac{4\,(f+g)^3\,(f-g)\,(f^4g^2+4\,g^3f^3-5\,f^2g^2+g^4f^2+4\,fg-1)}{(g^2+2\,fg-1)^2\,(f^2+2\,fg-1)^2}.$$

Since its numerator contains $f-g$ as a factor and $f+g$ is obviously never zero, we conclude that the proof will be completed provided we show that the long parenthesis

$$Z = f^4g^2+4\,g^3f^3-5\,f^2g^2+g^4f^2+4\,fg-1$$

in the numerator is always positive.

First note that the sum $\frac{A}{2}+\frac{B}{2}$ of halves of the angles is at most $\frac{\pi}{2}$ so that

$$\cot\left(\frac{A}{2}+\frac{B}{2}\right) = \frac{\cot(\frac{A}{2})\cot(\frac{B}{2})-1}{\cot(\frac{A}{2})+\cot(\frac{B}{2})} = \frac{f\,g-1}{f+g} > 0.$$

We conclude that $f\,g > 1$.

The first and the fourth term of Z together give

$$f^4g^2 + f^2g^4 = (f^2+g^2)(f\,g)^2 \geq 2(f\,g)(f\,g)^2 = 2(f\,g)^3$$

because $f^2+g^2 \geq 2fg$. If we introduce the notation $\vartheta = fg$ then

$$Z \geq 6\vartheta^3 - 5\vartheta^2 + 4\vartheta - 1.$$

Since $\vartheta > 1$ we can replace ϑ in the above cubic polynomial with $1+\vartheta$ and get $(3\,\vartheta+2)\,(2\,\vartheta^2+3\,\vartheta+2)$. This expression is always positive because the new variable ϑ is positive. This completes the proof.

Notice that the same could be obtained with the substitution $f = \frac{1+k}{g}$ for the positive real number k in the polynomial Z. Following the input

```
Collect[Extract[Q,4] /.  f->(1+k)/g, g];
```

the program Mathematica outputs

$$(1+k)^2\,g^2 + \frac{(1+k)^4}{g^2} + 2 + 6\,k + 7\,k^2 + 4\,k^3$$

which is obviously always positive. □

We continue with the problem 833 from [6] which is in the section about similarity of triangles.

Problem 6. Let r be the radius of the circle inscribed to a triangle ABC and let R be the radius of its circumscribed circle. Prove that $R \geq 2r$.



FIGURE 11. The diameter of the incircle is at most equal to the radius of the circumcircle.

Proof. The following proof has great similarity with the solution of the previous problem. Without loss of generality we can assume that the angles A and B of the triangle ABC are acute (i. e. less than $\frac{\pi}{2}$ radians) and that the vertices A, B and the center I of the incircle have the coordinates $(0, 0)$, $(r(f+g), 0)$ and (fr, r) for some real numbers $f > 1$, $g > 1$ and $r > 0$.

Our idea of the proof is first to determine the coordinates of the vertex C and the center O of the circumcircle. This will make it possible to compute the radius R of the circumcircle. Finally, we show that the difference $R - 2\,r$ is always positive except in the case of the equilateral triangle when it is zero.

Let J_a, J_b, J_c be projections of the center I of the incircle onto the sides of the triangle ABC. The point J_c has the coordinates $(f\,r, 0)$ while we get the coordinates of the J_a from the following system of the equations

```
sys:=Solve[{distance[tB,{p, q}]==distance[tB,tJc],
        distance[tI,{p, q}]==r},{p, q}];
```

where p and q are the wanted coordinates of the point J_a. This system has two solutions: the coordinates of the point J_c and the coordinates $\frac{\left(g^2 f+2\,g+f\right)r}{1+g^2}$ and $\frac{2rg^2}{1+g^2}$ of J_a. In a similar way we get the coordinates $\frac{f(f^2-1)r}{f^2+1}$ and $\frac{2\,f^2\,r}{f^2+1}$ of the point J_b.

```
tJa:={p,q} /.  Extract[sys, 2]
tJb:={p,q} /.  Extract[Solve[{distance[tA,{p, q}]==
  distance[tA,tJc], distance[tI,{p, q}]==r},{p, q}], 2]
```

The vertex C is the intersection $AJ_b \cap BJ_a$.

```
tC:=inter[line2[tA,tJb],line2[tB,tJa]];
```

The center O of the circumcircle and its radius R are given as solutions of the following system of equations:

```
tO:={p, q}; Solve[{distance[tA,tO]==R,
distance[tB,tO]==R, distance[tC,tO]==R},{p, q, R}];
```

From the two solutions of the system only the one where

$$R = \frac{r\,(1+g^2)\,(1+f^2)}{4(fg-1)}$$

is correct. In the second solution the radius R is negative which is not acceptable.

```
R:=r*(1+f^2)*(1+g^2)/4/(f*g-1);
M:=Collect[Extract[FS[R-2*r],2],f];
\[CapitalDelta]:=FS[Coefficient[M,f,1]^2-
      4*Coefficient[M,f,2]*Coefficient[M,f,0]]
```

The difference $R - 2\,r$ is equal to $\frac{M\,r}{4(f\,g-1)}$, where M is the quadratic trinomial

$$(g^2 + 1)f^2 - 8\,g\,f + g^2 + 9$$

in f. Its discriminant is $-4\,(-3 + g^2)^2$ which is always negative (so that $M > 0$ because the leading coefficient $g^2 + 1$ is positive) except when $g = \cot\frac{B}{2} = \sqrt{3}$ and $f = \sqrt{3}$ (i. e. the triangle ABC is equilateral) when $M = 0$. □

Next is the problem 312 from [6] which is in the second chapter on the perimeter and the area of circles.

Problem 7. Let T be a point inside the triangle ABC and let A_1, B_1, C_1 be interior points of the sides $\overline{BC}$, $\overline{CA}$, $\overline{AB}$. Let R_i for $i = 1, 2, 3, 4, 5, 6$ be radii of the circumcircles of the triangles AC_1T, C_1BT, BA_1T, A_1CT, CB_1T, B_1AT. Prove that $R_1\,R_3\,R_5 = R_2\,R_4\,R_6$.



FIGURE 12. The radii of the six circumcircles satisfy the relation $R_1\,R_3\,R_5 = R_2\,R_4\,R_6$.

Proof. Let us first define in Mathematica the function which associates to a given triple of points the radius of the circumcircle of the triangle whose vertices are these points.

```
bisector[a_, b_]:=perpen[midpoint[a,b],line2[a,b]];
CC[a_, b_, c_]:=inter[bisector[a,b],bisector[a,c]];
RC[a_, b_, c_]:=distance[a,CC[a, b, c]];
```

Let us now input the points A, B, C and T.

```
tA:={0, 0}; tB:={c, 0}; tC:={s, t}; tT:={p, q};
```

If $s \neq c$ then the position of a point A_1 on the line BC can be described by a real number u and the coordinates of this point are $\left(u,\, \frac{t(c-u)}{c-s}\right)$. We get this by requiring that the point with the coordinates $(u,\, z)$ lies on the line BC and then solve the condition with respect to z.

```
tA1:={u, z} /.  Solve[onlineQ[{u, z}, line2[tB, tC]]==0, z];
```

Similarly, if $s \neq 0$, then any point B_1 on the line CA has the coordinates $\left(v,\, \frac{t\,v}{s}\right)$ and any point C_1 on the line AB has the coordinates $(w,\, 0)$ for some real numbers v and w.

```
tA1:={u, t*(c-u)/(c-s)}; tB1:={v, t*v/s}; tC1:={w, 0};
```

If $s = c$ then any point A_2 on the line BC has the coordinates $(c,\, u)$ for some real number u. If $s = 0$ then any point B_2 on the line CA has the coordinates $(0,\, v)$ for some real number v.

```
tA2:={c, u}; tB2:={0, v};
```

Let us now define a function which computes the difference of the squares of the products of radii of the circumcircles for seven points in the plane.

```
FR[a_, b_, c_, d_, e_, f_,g_]:= FS[(RC[a,f,g]*RC[b,d,g]*
     RC[c,e,g])^2-(RC[f,b,g]*RC[d,c,g]*RC[e,a,g])^2];
```

It is now easy to check that the following values are zero:

```
     FR[tA, tB, tC, tA1, tB1, tC1, tT]
     s:=c; FR[tA, tB, tC, tA2, tB1, tC1, tT]
     s:=0, FR[tA, tB, tC, tA1, tB2, tC1, tT]
```

This completes the solution of the Problem 312 from [6] in the program Mathematica. □

Remark 1. It is clear from the above proof that we have never used the assumption that the point T is inside of the triangle ABC nor the assumption that the points A_1, B_1, C_1 are interior points of the sides $\overline{BC}$, $\overline{CA}$, $\overline{AB}$. In this way, using the computer, we succeeded to prove a more general statement.

The following example is the Problem 644 from the collection [6] which is in the section on the volume of the cylinder, cone, and ball.

Problem 8. On the bottom of the cylindrical container whose base has the diameter 15 cm there is a ball with the diameter 12 cm. The water is poured into the container up to the highest point of the ball. For how many cm will drop the level of the water when the ball is taken out?

Proof. Recall the formulas $V_B = \frac{4}{3}\left(\frac{D}{2}\right)^3 \pi$ for the volume of the ball with the diameter D and $V_C = \left(\frac{d}{2}\right)^2 h\,\pi$ for the volume of the cylinder of the height h whose base is a circle with the diameter d.



FIGURE 13. The cylindrical container with the ball inside.

In the program Mathematica these volume functions are defined as follows:

```
VB[d_]:=d^3*Pi/6; VC[d_,h_]:=d^2*h*Pi/4;
```

The volume of the water in the container is the difference of the volume of the cylinder (with the height equal to the diameter of the ball) and the volume of the ball:

```
Vwater:=VC[15, 12]-VB[12];
```

After the removal of the ball the water will fill in the cylindrical container whose base is the circle with the diameter of 15 cm and its height will be $12 - p$ cm where p is the required drop in the level of the water in the container. This drop p is found in the program Mathematica as follows:

```
Solve[Vwater==VC[15,12-p],p]
```

The solution is $p = 5.12$ cm. □

Another nice example is the Problem 963 from [6]. We assume again the knowledge of trigonometric functions.

Problem 9. A trapezium is circumscribed about the circle with the radius R. The chord that joins the touching points of the lateral sides has the length b and is parallel to the bases. Prove that the area of the trapezium is $\frac{8R^3}{b}$.



FIGURE 14. The area of the trapezium $ACEG$ is $\frac{8R^3}{b}$.

Proof. Select the rectangular coordinate system so that the circle k with the radius R inscribed to the trapezium $ACEG$ has the center in the origin and its parallel sides (bases) AC and EG touch k in the points $B(0, -R)$ and $F(0, R)$. Let the lateral sides CE and AG touch k in the points $D(R\cos\theta, R\sin\theta)$ and $H(R\cos\sigma, R\sin\sigma)$ for some angles θ and σ.

Let us first input into the program Mathematica the points O, B, F, D, H and the lines AC, EG.

```
tO:={0, 0}; tB:={0, -R}; tF:={0, R};
tD:={R Cos[\[Theta]], R Sin[\[Theta]]};
tH:={R Cos[\[Sigma]], R Sin[\[Sigma]]};
pAC:={0, 1, R}; pEG:={0, 1, -R};
```

Then we ask when will the chord DH joining the touching points D and H of the lateral sides be parallel with the bases.

```
parallelQ[line2[tD, tH], pAC]
```

The condition is $R(\sin\theta - \sin\sigma) = 0$ so that we must have $\sigma = \pi - \theta$. Hence, the trapezium $ACEG$ is equilateral and symmetrical with respect to the line BF. It suffices therefore to find the area only of the right half $BCEF$.

The line CE is the perpendicular in the point D to the line OD (the property of the tangent to the circle) and the points C and E are the intersections of the line CE with the lines AC and EG.

```
pCE:=perpen[tD, line2[tO, tD]];
```

```
tC:=inter[pAC, pCE]; tE:=inter[pEG, pCE];
```

The area of the right half $BCEF$ is the sum of the areas of the triangles BCE and BEF.

```
FS[area[tB, tC, tE]+area[tB, tE, tF]]
```

The program Mathematica will compute that this sum has the value $\frac{2R^2}{\cos\theta}$. Since $b = 2\,R\cos\theta$, we conclude that the wanted area of the trapezium $ACEG$ is indeed $\frac{8\,R^3}{b}$. □

Remark 2. In the book [6] there is the incorrect claim that the area of the trapezium is $\frac{4R^3}{b}$.

Using the approach from the solution of the Problem 13 (i. e., the Problem 1112 from [6]) it is possible to completely avoid the trigonometric functions. This solution we leave to the readers as an exercise.

We continue with the solution of the Problem 1026 from [6].

Problem 10. Prove that in every regular heptagon $A_1A_2A_3A_4A_5A_6A_7$ the following equality holds:

$$\frac{1}{|A_1A_2|} = \frac{1}{|A_1A_3|} + \frac{1}{|A_1A_4|}.$$



FIGURE 15. The relation $\frac{1}{|A_1A_2|} = \frac{1}{|A_1A_3|} + \frac{1}{|A_1A_4|}$ holds in every regular heptagon $A_1A_2A_3A_4A_5A_6A_7$.

Proof. Choose the coordinate system so that the circle k with the center at the origin and with the radius R is circumscribed to the heptagon $A_1A_2A_3A_4A_5A_6A_7$. We can assume that the vertex A_1 has the coordinates $(R,\,0)$. The other relevant vertices have the coordinates $A_2\left(R\cos\frac{2\pi}{7},\,R\sin\frac{2\pi}{7}\right)$, $A_3\left(R\cos\frac{4\pi}{7},\,R\sin\frac{4\pi}{7}\right)$, $A_4\left(R\cos\frac{6\pi}{7},\,R\sin\frac{6\pi}{7}\right)$.

Let us input these points into the program Mathematica:

```
tA1:={R, 0};
tA2:={R Cos[2 Pi/7], R Sin[2 Pi/7]};
tA3:={R Cos[4 Pi/7], R Sin[4 Pi/7]};
tA4:={R Cos[6 Pi/7], R Sin[6 Pi/7]};
```

In order to check the above relation among the reciprocal values we must type into the program Mathematica the following:

```
FullSimplify[Numerator[Together[1/distance[tA1, tA2]
-1/distance[tA1, tA3]-1/distance[tA1, tA4]]], R>0]
```

For few seconds the computer will output the value zero which proves that the statement in the problem holds. □

Remark 3. Several other interesting properties of the regular heptagon proved in the program Maple V are described in the article [5].

Next is the Problem 1084 from the section eight of the collection [6].

Problem 11. The projections of the legs of the right triangle onto the hypothenuse have lengths $\frac{18}{5}$, $\frac{32}{5}$. Find the radius of the circle inscribed into this triangle?



FIGURE 16. The projections AD and DB of the legs are known. We look for the inradius r.

Proof. Select the rectangular coordinate system so that its origin is the vertex C of the right triangle and its legs are on the coordinate axes. We can assume that the remaining vertices A and B have the coordinates $(0, b)$ and $(a, 0)$, for some positive real numbers a and b.

In the program Mathematica these points are input as follows:

```
tC:={0, 0}; tA:={0, b}; tB:={a, 0};
```

Then we find the projection D of the vertex C onto the hypotenuse

AB.

```
tD:=project[tC, line2[tA, tB]];
```

The values for the variables a and b can be determined from the information that $|AD| = \frac{18}{5}$ and $|BD| = \frac{32}{5}$.

```
Solve[{distance[tA,tD]==18/5,
      distance[tB,tD]==32/5},{a, b}];
```

There are eight solutions (four real and four complex) but only one when $a = 8$ and $b = 6$ is acceptable. Hence, this right triangle has sides 8, 6, 10 (that are twice as long as the sides of the standard (Egyptian) right triangle with sided 4, 3, 5) so that its inscribed circle has the radius $r = 2$.

This could also be seen by asking that the center I of the inscribed circle with the coordinates (r, r) is at the distance r from the line AB.

```
a:=8; b:=6; tI:={r, r}; Solve[distance[tI,
  project[tI, line2[tA, tB]]]==r, r];
```

From the two solutions $r = 2$ and $r = 12$ only the first satisfies the conditions of the problem. The second solution gives the radius of the corresponding excircle. □

Now we consider the Problem 1103 again from the collection [6].

Problem 12. Two sides of the triangle have the length 6 cm and 8 cm. The medians of these sides are perpendicular. Find the third side of this triangle.



FIGURE 17. Find the side c if the sides a and b are known and the medians AT_a and BT_b are perpendicular.

Proof. Let the triangle ABC be embedded into the rectangular coordinate system so that $A(0,\,0)$, $B(c,\,0)$, and $C(u,\,v)$ for positive real numbers c and v and for a real number u.

In the program Mathematica these points and the centroid T are input as follows:

```
tA:={0, 0}; tB:={c, 0}; tC:={u, v}; tT:=centroid[tA,tB,tC];
```

Since the medians of the vertices A and B are perpendicular, ABT is the right triangle and $c^2 = |AB|^2 = |AT|^2 + |BT|^2$ by the Pythagorean theorem. On the other hand $|BC| = 6$ and $|AC| = 8$. If we ask the program Mathematica to solve this system of three equations in the variables c, u, and v with the input

```
    Solve[{distance[tB,tC]==6, distance[tA,tC]==8,
c^2==distance[tA,tT]^2+distance[tB,tT]^2},{c, u, v}]
```

it will respond with two solutions. Only the one where $c = 2\sqrt{5}$ cm is correct. □

Our next example is the Problem 1112 from [6].

Problem 13. A circle is inscribed into a trapezium. Prove that the ratio of the areas of the circle and the trapezium is equal to the ratio of their perimeters.



FIGURE 18. The ratios of the areas and of the perimeters of the circle and the circumscribed trapezium are equal.

Proof. Choose the rectangular coordinate system so that the circle k with the radius R which is inscribed to the trapezium $ACEG$ has the center in the origin while its parallel sides (bases) AC and EG touch k in the points $B(0,\,-R)$ and $F(0,\,R)$. Let the vertices A and C have the coordinates $(-u,\,-R)$ and $(v,\,-R)$ for positive real numbers u and

v. Let the lateral sides CE and AG touch k in the points D and H. Our first goal is to find the coordinates of these points and then the coordinates of the vertices E and G.

Let us first input into the program Mathematica the points O, B, F, A, C and the lines AC, EG.

```
tO:={0, 0}; tB:={0, -R}; tF:={0, R};
tA:={-u, -R}; tC:={v, -R};
pAC:={0, 1, R}; pEG:={0, 1, -R};
```

Assume that the point H has the coordinates (p, q). They must satisfy two conditions. The first is $p^2 + q^2 = R^2$ i. e. that the point H lies on the circle k. The second condition is that the distance from A to H is equal to u because the lines AB and AH are tangents through the point A onto the circle k.

```
H:=Solve[{p^2+q^2==R^2, distance[{p, q},tA]==u}, {p, q}]
tH:={p, q} /. H;
```

In a similar way we can determine the coordinates of the point D.

```
K:=Solve[{p^2+q^2==R^2, distance[{p, q},tC]==v}, {p, q}]
tD:={p, q} /. K;
```

The vertices E and G are the intersections of the line EG with the lines CD and AH, respectively.

```
pAH:=line2[tA, tH]; pCD:=line2[tC, tD];
tE:=inter[pEG, pCD]; tG:=inter[pEG, pAH];
```

The first coordinates of the points E and G are $\frac{R^2}{v}$ and $-\frac{R^2}{u}$. Hence, the perimeter O_{ACEG} of the trapezium $ACEG$ is $2(u + v + \frac{R^2}{u} + \frac{R^2}{v})$. Its area P_{ACEG} is

```
FS[area[tA, tC, tE]+area[tA, tE, tG]]
```

equal to $\frac{R(u+v)(u\,v+R^2)}{u\,v}$. Now it is easy to check that

$$\frac{2\,R\,\pi}{O_{ACEG}} = \frac{R^2\,\pi}{P_{ACEG}}.$$

□

Remark 4. In [6] there are no solutions for the Problem 1112.

The next example is the Problem 1139 from [6].

Problem 14. Prove that if the angle bisector of a triangle is also the bisector of the angle determined by the altitude and the median, then this triangle is right.



FIGURE 19. The triangle is either isosceles or right if the angle bisector is also the bisector of the angle between the altitude and the median.

Proof. Let us choose the rectangular coordinate system so that the points $A(0, 0)$, $B((f+g)r, 0)$, $C\left(\frac{rg\left(f^2-1\right)}{fg-1}, \frac{2fgr}{fg-1}\right)$ are the vertices of the triangle and the center of its inscribed circle is the point $I(fr, r)$, where f and g are cotangents of $\frac{A}{2}$ and $\frac{B}{2}$ and r is the radius of the incircle.

We shall first input into the program Mathematica the points A, B, the midpoint C_g of the segment AB, the points C, I and the feet C_h of the altitude of the vertex C on the line AB.

```
tA:={0, 0}; tB:={r*(f+g), 0}; tCg:=midpoint[tA,tB];
tC:={r*g*(f^2-1)/(f*g-1), 2*f*g*r/(f*g-1)};
tI:={f*r, r}; tCh:=project[tC,line2[tA,tB]];
```

In order that the bisector of the angle C (i. e. the line CI) is the bisector of the angle between the altitude (i. e. the line CC_h) and the median (i. e. the line CC_g) it is necessary and sufficient that the segments II_h and II_g have the same length, where I_h and I_g are the projections of the point I onto the lines CC_h and CC_g.

```
tIh:=project[tI,line2[tC,tCh]];
tIg:=project[tI,line2[tC,tCg]];
IZ:=FS[distance[tI,tIg]^2-distance[tI,tIh]^2]
```

The program Mathematica reports that the expression IZ is equal

$$\frac{r^2\,(f-g)^2\,(fg+g+f-1)\,(fg-g-f-1)\,(fg+1)^2\,(fg-1)^{-2}}{(12\,f^2g^2+g^2f^4-2\,f^3g^3+g^4f^2+2\,f^3g+2\,fg^3+f^2-2\,fg+g^2)}.$$

Hence, it will be zero if and only if $f = g$ (i. e. $|BC| = |CA|$ so that the triangle ABC is isosceles) or

$$(f g + g + f - 1)(f g - g - f - 1) = 0$$

which is the condition for the lines BC and CA to be perpendicular (i. e. that the angle C has 90 degrees and the triangle ABC is right).

```
perpenQ[line2[tB, tC], line2[tC, tA]]
```
□

Remark 5. In the collection [6] the possibility that the triangle ABC is isosceles is absent.

Our final example is the Problem 1152 from [6].

Problem 15. Let different points A and B be given and let the point T be outside the line AB. Through the point T construct the line m so that the ratio of the distances of the points A and B to the line m is $2 : 3$.



FIGURE 20. There are two lines a and b through the point T such that the ratios of their distances to the points A and B are $\frac{2}{3}$.

Proof. Choose the rectangular coordinate system so that the given points are $A(0, 0)$, $B(c, 0)$, and $T(p, q)$ for real numbers c, p, q. Let the line m has the equation $u\,x + v\,y + w = 0$ for some real numbers u, v, w. In order that it goes through the point T the free term w must be equal to $-u\,p - v\,q$.

Let us input into the program Mathematica the points A, B, T and the line m.

```
tA:={0, 0}; tB:={c, 0}; tT:={p, q}; pm:={u, v, -u*p-v*q};
```

Let A_m and B_m be the projections of the points A and B onto the line m.

```
tAm:=project[tA, pm]; tBm:=project[tB, pm];
```

By the requirement of the problem the quotient $\frac{|AA_m|}{|BB_m|}$ is equal to $\frac{2}{3}$. Notice that the expression

```
IZ:=FS[distance[tA,tAm]^2/distance[tB,tBm]^2-4/9]
```

has as the numerator the product $(5\,up + 5\,vq - 2\,uc)\,(up + vq + 2\,uc)$. Hence, there are two possibilities $q = \frac{-u(2c+p)}{q}$ and $q = \frac{u(2c-5p)}{q}$. They give lines $q\,x - (2c+p)y + 2qc = 0$ and $5q\,x + (2c-5p)y - 2qc = 0$ as solutions of the problem. Even though we know the solutions the question remains how to construct them. But, it is simple to see that they intersect the line AB in the points $C(-2c,\,0)$ and $D(\frac{2}{5}c,\,0)$ and these are easily constructed. □

Remark 6. In the collection [6] there are no solutions for the problem 1152.

References

[1] M. Bator, Z. Čerin and M. Ćulav, *Analitička geometrija ravnine računalom*, Matematičko-fizički list, (in Croatian, to appear).

[2] M. Bator, Z. Čerin and M. Ćulav, *Analitička geometrija ravnine i Mathematica*, PrimMath 2003, (in Croatian, preprint).

[3] Zvonko Čerin, *Šest zadataka riješenih programom Maple V*, Matematičko-fizički list, (in Croatian, to appear).

[4] Zvonko Čerin, *Još šest zadataka riješenih programom Maple V*, Matematičko-fizički list, (in Croatian, to appear).

[5] Zvonko Čerin, *Geometrija pravilnog sedmerokuta*, Poučak (and/or PlayMath), (in Croatian, to appear).

[6] Boris Pavković and Darko Veljan, *Matematika 1 – Zbirka zadataka s uputama i rješenjima za prvi razred srednjih škola (14. dopunjenjo izdanje)*, Školska knjiga, Zagreb, 1999. (in Croatian).

Odranska 8, 10000 Zagreb, Croatia, Europe
E-mail address: batormaja@yahoo.com

Kopernikova 7, 10020 Zagreb, Croatia, Europe
E-mail address: cerin@math.hr

Lhotkina 3, 10000 Zagreb, Croatia, Europe
E-mail address: mickey@student.math.hr